CHAPTER

13 Scripting BBEdit

BBEdit offers access to nearly all of its features and commands via AppleScript. This chapter provides a brief overview of AppleScript, discusses BBEdit's scripting model, and explains how you can use scripts within BBEdit.

An excellent way to learn how to script BBEdit is to look at the scripts others have written for it, or to turn on recording in your script editor while you perform actions in BBEdit. The BBEdit Talk discussion group is also a good resource for learning more about scripting.

```
https://groups.google.com/g/bbedit
```

IMPORTANT Regardless of whether you are new to scripting BBEdit or are familiar with scripting previous versions, we strongly recommend that you carefully review the sections "BBEdit and AppleScript" and "Working with Scripts" in this chapter.

In this chapter

AppleScript Overview

If you are familiar with AppleScript, you should have little difficulty scripting BBEdit. It has a robust and highly flexible object model. If you do not know much about scripting, though, read on for an introduction to the necessary concepts.

About AppleScript

AppleScript is an English-like language which you can use to write scripts that automate the actions of applications, and exchange data between applications. Although AppleScripts can manipulate applications' user interfaces by taking advantage of the system's GUI Scripting capability, this is not their primary function. Rather, scripts talk directly to a application's internals, bypassing its user interface and interacting directly with its data and capabilities.

If you want to insert some text into a document, emulating a user typing into an editing window is not the most efficient way of accomplishing this. With AppleScript, you just tell the application to insert the text directly. If you want the application to save the frontmost document, you need not mime choosing Save from the File menu, but rather just tell the application to save its frontmost document.

Note AppleScript is actually a specific language which resides atop the general Open Scripting Architecture (OSA) provided by macOS. Although AppleScript is by far the most common OSA language, there are others, including a JavaScript variant. All OSA languages are capable of accomplishing similar things, although the actual commands used differ from one language to the next. In this chapter, we will focus exclusively on AppleScript, since it is the standard scripting language, but you should bear in mind that there are other options.

Scriptable Applications and Apple Events

Since AppleScripts must have direct access to an application's internal data structures, any application that will be used in an AppleScript must be designed to allow this access. We say such applications are *scriptable*. BBEdit is scriptable, as are many, many other programs. However, it is important to note that not *every* application is scriptable, and AppleScripts are not the best solution for automating applications that are not.

What goes on in an application that is scriptable? The foundation of AppleScript is something called the *Apple Event*. Macintosh applications are designed around an event loop; they go around in circles waiting for you, the esteemed user, to do something (choose a menu command, press some keys, and so on). These actions are passed to the application by the operating system in the form of an *event*. The application decodes the event to figure out what you did, and then performs an appropriate operation. After an event has been handled, the application goes back to waiting for another one. (At this point, the Mac OS may decide to give some time to another application on your computer.)

Apple Events are special events that applications send to each other, enabling a feature called *inter-application communication* (IAC). (It's a mouthful, but it just means applications can talk to each other.) Apple Events are also the way AppleScripts tell applications what to do, and which data to retrieve. So to be scriptable, an application must first support Apple Events.

Apple Events in their naked form are raw and cryptic things—bits of hieroglyphics only a programmer could love. So a scriptable application also has a *scripting dictionary*. The scripting dictionary tells any application that lets you write AppleScripts, such as the standard Script Editor, the English-like equivalent for each Apple Event and each event's parameters.

It is important to note that because Apple Events were originally designed to allow applications to communicate with each other, AppleScripts automatically inherit the ability to talk to more than one application. It is common in the publishing industry, for instance, to write scripts that obtain product information from a FileMaker Pro database and insert it into an InDesign file. This integration is one of the Macintosh's primary strengths.

You use AppleScript's *tell* verb to indicate which application you are talking to. If you are only sending one command, you can write it on one line, like this:

```
tell application "BBEdit" to count text documents
```

If you are sending several commands to the same application, it is more convenient to write it this way:

```
tell application "BBEdit"
    count text documents
    repeat with x from 1 to the result
        save text document x
    end repeat
end tell
```

The Script Editor automatically indents the lines inside the *tell* block for you so you can more easily follow the organization of the script.

Reading an AppleScript Dictionary

To display an application's AppleScript dictionary, you can simply drag that application onto the Script Editor icon, or use the Script Editor's Open Dictionary command. As we noted earlier, all scriptable applications include a dictionary that tells AppleScript how to convert English-like commands into the Apple Events actually expected by the application. The Script Editor uses this same information to display a sort of "vocabulary guide" that helps you write your scripts.

We will naturally use BBEdit's dictionary, shown below, to illustrate how to read a dictionary.



(You will probably want to make the window bigger if you have room on your screen.)

Down the left side is a list of every event and object supported by the application. An event is a verb—it tells the application what to do. A class is a noun: a piece of data, or a structured collection of data, inside the program. In BBEdit, for instance, classes are things like files, windows, the clipboard, browsers, and so on.

Suites

The first thing you will notice is that the events and classes are divided into *suites.* A suite is just a collection of related events and classes. Apple, for instance, has decreed that all applications should support particular events, which together are called the Required Suite. Another Apple-defined suite is the Standard Suite: if an application offers certain functions which Apple considers to be common, it should use these standard terms, so that scripters do not need to learn a new term for each application they work with. After that, it is a free-for-all—each developer is free to organize their events and classes however they think best.

In addition to the Required and Standard suites, BBEdit has a Miscellaneous suite, a BBEdit Suite, a Text suite, and an HTML Scripting suite.

Within each suite, events—verbs—are displayed in normal text, while classes—nouns—are italicized. Most commands sent to BBEdit will start with one of the verbs. (In some cases, *get* might be implied.)

Events

Let's look more closely at one of the events—*Save* is a good one to start with. It is shown below.



The right side of the window shows the syntax of the selected event, as well as a brief description of its function. The boldface words are keywords; they must be included exactly as shown or the script will not compile. The normal text tells you what kind of information goes after each keyword. For example, after *save* you must give a reference; the italicized comment next to that line indicates that it is a reference to the window to be saved. In other words, some window object, which in BBEdit would be *window 1* for the frontmost window, or *window "Text File"* if you want to specify a window by name. (we will show you how to figure all that out in a moment—you have to look at the window class's dictionary entry.)

Anything in square brackets is optional. Most of the rest of the *save* event is optional, in fact. The basic event just saves the frontmost window to the same file from which it was opened. However, you can also optionally include the word *to* followed by a file reference. (You specify a file simply by using the word *file* followed by the path name of the file, as in *file "Hard Disk:Users:BBSW:Documents:My file".)* If you specify a file to save the window to, the text will be saved into that file instead of the file it came from—like using Save As instead of Save.

The last three optional parts of the *save* event are denoted as boolean. That means they take either a true or a false value. In AppleScript, there are a couple of different ways to specify boolean values. You can write *saving as stationery true* to tell BBEdit to save the file as a stationery document. Or you can write *with saving as stationery.* You will notice that the last two parameters default to true if you do not specify them as false. To do that, you would use *add to recent list false* or *without add to recent list*. Whichever way you write it, you will notice that when you compile the script, AppleScript rewrites it using "with" or "without". Since that is the syntax AppleScript seems to like best, that is probably the one you should get used to thinking in.

Let's take a look at another one: the prosaic *get.* Select *get* from BBEdit's dictionary listing and take a quick look at its class definition.You use *get* to retrieve information from an application. You must specify a reference to the object you want to retrieve, and you can specify a *coercion*—a condition that tells AppleScript to treat one type of data as if it were another—by adding the *as* clause. However, after that is the *Result:* line, which we have not seen before. This line tells you what type of value the command returns. (This value is placed in the AppleScript system variable called *the result.) Get* can retrieve any kind of object, so it can return anything, as indicated here. Other events might return a specific type of result, or none at all. *(Save* did not have a *Result:* line in its dictionary entry, which means it does not return a result.)

Classes and the Class Hierarchy

Let's look at a typical class definition: *window* will do nicely. It is in the BBEdit Suite, toward the bottom.



All windows in BBEdit belong to this class. A class defines a particular kind of object; a particular example of an object belonging to the class is said to be an instance of that class, or just an object of that class. So here we are looking at the class itself; each individual window object has all these properties.

After a tag line that tells you about the class ("an open window") comes the plural form. AppleScript lets you refer to windows either singly or as a group, so it needs to know what the plural of every term is. For example, try this little script:

```
tell application "BBEdit" to count windows
```

The result of this script is the total number of window objects currently displayed by BBEdit.

After the plural form comes a list of properties. Some objects do not have properties—for example, a string—but many applications do. An object's properties are merely a collection of data that describes that particular object. For example, as you look down the list of window properties, you will see that every window has a name, every window has a position, every window has bounds (the area of the screen it covers), and so on.

The first item on the list, though, is <*inheritance*> *item*. This tells you that a window is a kind of item, and that it therefore has all the properties of an item. Take a quick look at *item's* class definition, shown below.



You will see three properties: *properties, ID*, and *container*. The first entry *properties* is a record containing all the object's properties. In other words, because a window is an item, it has, in addition to all its listed properties, another property which returns all the other properties as a record—a single piece of data that can be stored in a variable. Every class in BBEdit is part of a hierarchy with the *item* class at the top, so every object in BBEdit "inherits" the *properties* property. This catch-all property can be handy for making exact duplicates of objects, among other uses.

You may realize that BBEdit has several kinds of windows; you can see their classes listed in the dictionary: differences window, disk browser window, project window, text window, tool window, and the like. Let's look at *text window:*



You can see that a text window inherits all the properties of the *window* class. And, since the *window* class inherits all the properties of the *item* class, this means that the *text window* class also has the *properties* property defined by the *item* class.

To make explicit what you might have already gathered, classes in AppleScript form a hierarchy. That is, classes can be based on other classes. Such a class is called a *subclass,* and the class on which a subclass is based is referred to as its *parent class.* (In AppleScript, classes can only have one parent. Multiple inheritance is a feature found in more complex languages.)

The idea of a class hierarchy makes it easier for us to add new features to BBEdit, since when we want to create a new kind of window, half the work is already done. However, when scripting, you may need to flip back and forth between two or more class definitions to find all the properties of the object you are working with. (This is, technically speaking, a limitation of Apple's Script Editor. There is no reason the inherited properties could not automatically be included in a subclass listing by a smarter editor, for example, Script Debugger, which does this.)

Now that we have the class hierarchy under control, let's look at the properties themselves more closely. we will stick with the *text window* class at this point.

Properties of an object are referred to using the preposition *of.* For example, the following line of script returns the font of the frontmost text window.

```
tell application "BBEdit" to get display font of
text window 1
```

Note In this specific example, you can just write *get display font of window 1*. AppleScript will figure out that window 1 is more specifically a text window, and therefore has a *display font* property, even though the generic *window* class does not have any such property. All the properties of the object are available even if you did not use its specific class name. However, in most cases, you should specify exactly the object you want; this distinction is especially important when dealing with text documents (content) versus text windows (display elements).

You can set the properties using the *set* event, like so:

```
tell application "BBEdit" to set display font of text
window 1 to "Courier New"
```

Let's go back to the *window* class for a moment. Most of the properties of this class are marked with the abbreviation *[r/o].* That stands for Read-Only. In other words, you can only *get* these properties, not *set* them.

Recordable Applications

Once an application accepts Apple Events, it actually makes a good deal of sense for an application to be designed in two parts: the user interface that you see, and the "engine" that does all the work. (An application designed this way is sometimes said to be *factored*.) The user interface then communicates with the engine via Apple Events.

The design of the Apple Event system makes it possible to "record" events into a script. This feature not only lets you automate frequently performed tasks with little hassle, it also can be an enormous aid in writing larger and more complicated scripts, because the application tells you what events and objects to use for the kind of task you record.

Because of the important recording functionality they enable, applications that have been factored and use Apple Events to let the two halves communicate are said to be *recordable*. It is important to note that not all scriptable applications are recordable.

Saving Scripts

Any AppleScript can be saved in what's called a *compiled script file*. A compiled script file contains the actual Apple Events; by generating these events when you save the file, the operating system does not have to convert your English-like commands into events each time you run the script, which means it loads faster. When double-clicked in the Finder, a compiled script file automatically opens in the Script Editor, where it can be run. A script can also be saved as a stand-alone application, or *applet*, in which case double-clicking the script's Finder icon automatically runs the script. Both types of files can be saved with or without the English-like *source code;* if you save it without the source code, other users you give the script to will not be able to make any changes to it (of course, you should also keep a copy of the script *with* the source for yourself).

Using Scripts with Applications

Although you can place a script applet in the global Scripts menu, or in any folder, and use it any time you need it, many applications (including BBEdit) provide a special menu that lets you launch compiled scripts intended specifically for use with that one application. Since you do not have to save them as applets, they take up less disk space and launch more quickly. They also show up only in the application you use them with, rather than cluttering your global Scripts menu.

Some applications go even further, allowing you to define scripts to be run when certain things happen in the program. For example, an application might let you define a script to be executed when the user chooses *any* menu item. The script might then perform some pre-processing, and then exit by telling the application whether to continue with the menu command or to cancel it. As a simple example, a script might check to see what printer is selected when the user chooses the Print command. If it is the expensive color dye-sublimation printer, on which printing a page costs several dollars, the script could remind the user of that fact and confirm their intention (through an alert) before continuing with the print operation.

An application that supports such a feature (or any method of integrating user-written scripts seamlessly into its user interface) is said to be *attachable,* because the scripts become "attached" to the features of the program. (BBEdit is attachable; more details about using this feature are provided later in this chapter.)

Scripting Resources

Covering all the details you might need to write your own AppleScripts is not something we can reasonably do in this manual. AppleScript, despite its deceptively simple English-like syntax, is a sophisticated object-oriented language with many subtleties. For this reason, we suggest you consult supplemental documentation and resources if you are a beginning scripter.

A good place to start is with someone else's script: find a script that does *almost* what you want it to and repurpose it. Even if you cannot find a script that does anything close to what you want, reading others' scripts is a good way to learn how AppleScript "thinks" and how BBEdit's particular AppleScript implementation behaves.

In addition to the basic AppleScript documentation included with the system, you may find the following resources useful in your quest to understand scripting.

Books

AppleScript: The Definitive Guide (Second Edition), Matt Neuberg. O'Reilly and Associates, 2006. ISBN: 0-596-10211-9

Discussion Groups

BBEdit Talk

`https://groups.google.com/g/bbedit`
The BBEdit Talk discussion group is an excellent place to ask BBEdit-specific scripting questions.

Mac Scripting

`http://listserv.dartmouth.edu/scripts/wa.exe?A0=MACSCRPT`
Unofficial list covers AppleScript and other Macintosh scripting languages, with occasional forays into peripheral topics.

Websites

AppleScript: The Language of Automation

`http://www.macosxautomation.com/applescript/`

MacScripter.Net

`http://macscripter.net/`
A good selection of AppleScript-related news and topics, including the "AppleScript FAQ" and discussion forums.

ScriptWeb

`http://www.scriptweb.com/`
This site covers all scripting languages, not just AppleScript. Also, it has an extensive directory of scripting additions.

Software

Script Debugger

`https://www.latenightsw.com/`

Despite its name, Script Debugger is more than a debugger; it is actually an enhanced replacement for Apple's Script Editor, featuring variable monitoring, step/trace debugging, an object browser for an application's objects, and much more.

Using AppleScripts in BBEdit

BBEdit has been scriptable for years, and we have continually worked to refine its level of scripting support. In addition to providing extensive script access to its commands and data, BBEdit is both attachable *and* recordable.

This section describes how you can create and employ AppleScripts within BBEdit via recording and BBEdit's various scripting facilities, while the following section covers BBEdit's scripting commands and other issues related to preparing scripts for use.

Recording Actions within BBEdit

Any language is easier to read than to write, easier to understand than to speak. AppleScript is no different. That's because, even though all the commands it uses are English words arranged in ways that more or less make grammatical sense, you still have to know (or find out from the application's dictionary) exactly which words to use, and what order they should go in. But it is easy to get started making scripts by recording them.

First, launch both BBEdit and the Script Editor.

When you launch the Script Editor, a new, blank script window appears. Click the Record button, circled in the illustration below.







Now switch to BBEdit and perform your task. Remember that the Script Editor is recording *everything* you do in every recordable application you are running, not just BBEdit. If you do something in the Finder, for instance, that will get recorded too. Since almost everything you do is recorded, remember that if you make an error, and then Undo it, your recorded script will faithfully make the same mistake and undo it when you run it later. It will be possible to fix minor errors later, but things always go more smoothly if you do not make any mistakes, so take your time and try to do it right the first time.

Now switch back to the Script Editor and click the Stop button. After a brief pause, your script is compiled and ready for use. Try clicking the Run button to see it work. (It might not work correctly. If you recorded a search and replace operation changing every "cat" to "dog", you already changed the document while recording the script, and of course the script will not do anything when you run it.)

Finally, save the script in the BBEdit Scripts folder so that it shows up in BBEdit's script menu. Choose Save As from the File menu, and then use the Script Editor's Save dialog to put the script in your BBEdit Scripts folder. Now try selecting it from the script menu in BBEdit.

The Scripts Menu

The Scripts menu (left) in BBEdit's menu bar contains several commands. It also lists all AppleScripts (as well as Automator actions, text factories, and Unix scripts) present in the Scripts folder within BBEdit's application support folder, providing a quick way to access frequently used scripts. You can place scripts within subfolders (up to 4 levels deep) of the Scripts folder to organize them.

Note AppleScripts written for use in as BBEdit filters or scripts should be saved as compiled (data fork) script files, not script applications.

In addition to the list of available scripts, the Scripts menu provides the following commands.

Open Script Editor

Choose this item to switch to the system's default AppleScript editor. If the script editor is not running, BBEdit launches it.

Open Scripting Dictionary

Choose this item to switch to your preferred AppleScript editor and open BBEdit's scripting dictionary for viewing. If the script editor is not running, BBEdit launches it.

Open Scripts Folder

Choose this item to open the Scripts folder which is located within BBEdit's application support folder. (See "Scripts" on page 36.)

Running and Editing Scripts

Choose the item corresponding to any script to run that script. To edit a script, you may either open the script file directly from within the "Scripts" subfolder of BBEdit's application support folder, or select the desired script in the Scripts palette and click the "Edit..." (pencil) button.

The Scripts Palette

The Scripts command, located in the Palettes submenu of the Window menu, opens a palette listing all available scripts. Names that are too long to fit within the width of the window are truncated with ellipses (…).

"Hovering" the mouse over such a truncated name displays a tool tip showing the full name. If you hold down the Option key, the tool tip will appear instantly, with no hovering delay. Names that fit entirely within the window without truncation do not display a tool tip.

Organizing Scripts

Items in the Scripts menu and Scripts palette are displayed in alphabetical order by default, but you can force them to appear in any desired order by including any two characters followed by a right parenthesis at the beginning of their name. (For example "00)Save All" would sort before "01)Close All.") For names of this form, the first three characters are not displayed in the window.

You can also insert a divider into the Scripts menu by including an empty folder whose name ends with the string "–***". (The folder can be named anything, so it sorts where you want it.)

Note Any dividers you add will appear in the Scripts menu, but not the Scripts palette.

Attaching Scripts to Menu Items

BBEdit lets you attach scripts to menu items. By this, we mean that you can write scripts that BBEdit automatically calls before or after performing a menu command. For example, if you want BBEdit's Open from FTP/SFTP Server command to launch your favorite FTP client, you can simply attach a script to that menu item. Scripts can return a value that tells BBEdit whether to continue with the command that was selected, or to cancel the operation (in which case only the script is executed).

Scripts attached to BBEdit menu items must be stored in the Menu Scripts folder of BBEdit's application support folder. These files should be compiled scripts, not script applications. Scripts are named to indicate which menu item they go with: first the name of the menu (or the submenu) upon which the item is immediately located, then a bullet "•" (Option-8) character, then the name of the menu item. For example, to attach a script to the Open from FTP/SFTP Server menu item, you would name it "File•Open from FTP/SFTP Server", while to attach a script to the New Document menu item, you would name it "New•Text Document".

Some of BBEdit's menus have icons rather than names. BBEdit uses the following names for its icon menus: "#!" [the 'Shebang' menu], "Compiler", and "Scripts". Furthermore, the New With Stationery submenu is named "Stationery" for purposes of attachability.

When you choose a menu command which has an attached script, BBEdit will pass the menu name and command (item) name to the script's MenuSelect handler, if it has one. If the script contains no MenuSelect handler, BBEdit executes the script's run handler.

The script's MenuSelect handler can tell BBEdit to skip performing the chosen command by returning "true", or have it continue on and perform the command by returning "false". If MenuSelect returns "false", BBEdit will call the script's PostMenuSelect handler, if it has one, after it performs the menu command.

Here is a simple example, which adds a confirmation dialog to the Save command (addressed as "File•Save"). Note that we test the menu and item names to make sure the script is attached to the Save command—if it is attached to some other command, it does nothing.

```
on menuselect(menuName, itemName)
  if menuName = "File" and itemName = "Save" then
    set weHandledCommand to true
    display dialog "Are you sure you want to save?" ¬
    buttons {"No", "Save"} default button 2
    if button returned of the result is "Save" then
    -- the application should do its work
    set weHandledCommand to false
    else
    -- we handled the command, app does no work,
    -- postmenuselect doesn't get called
    display dialog "The document was not saved." ¬
      buttons {"OK"} default button 1
    end if
    return weHandledCommand
  end if
end menuselect

on postmenuselect(menuName, itemName)
  -- this is called after the application has processed
  -- the command
  display dialog "The document was saved." ¬
    buttons {"OK"} default button 1
end postmenuselect
```

Attaching Scripts to Events

IMPORTANT BBEdit offers script attachability, which means you are not limited to menu commands but can attach scripts directly to the app which will take effect whenever selected application and/or document events occur.

To access these events, your attachment scripts must contain function names which correspond to the names of the events' attachment points. Except when otherwise noted, all of the following considerations apply:

- Every function takes a single argument which is a reference to the object in question: the application for application entry points, or the document being opened/closed/saved/etc for document entry points.
- Any function associated with an attachment point whose name contains 'should' must return a Boolean result: 'true' or 'false'. If it returns 'true', the operation will continue. If it returns 'false' or throws an error (see below) then the operation will be cancelled. So, for example, 'applicationShouldQuit' returning 'true' will allow the application to quit; returning 'false' will not.

- If an attachment script causes a scripting error and does not handle it within the script itself, BBEdit will report the error. In the case of functions which are used to allow a ‘should’ action, this will prevent the action from occurring.

Here are the available attachment points:

Application attachment points

- *applicationDidFinishLaunching*: called when the application has completed startup.
- *applicationShouldQuit*: called when you choose the Quit (or the application receives a ‘quit’ event for any other reason).
- *applicationDidQuit*: called when the application has finished shutting down and is about to exit.
- *applicationDidSwitchIn*: called when BBEdit has been brought to the foreground.
- *applicationWillSwitchOut*: called when BBEdit is being put into the background. You could use this (for example) to save outstanding changes to the front document.

NOTE In BBEdit 13.1, the attached script will run **before** the app goes into the background, rather than **after** (as in previous versions). You should also avoid using ‘show dialog’ or similar verbs during *applicationWillSwitchOut*, because that will leave the resulting item on screen until you switch back to BBEdit (and in the event you have also defined an attachment for *applicationDidSwitchIn*, that will likewise run so you'll **really** be in the soup).

Document attachment points

- *documentDidOpen*: called when a document has been opened and is ready for use. (Since BBEdit supports multiple types of documents, your script should allow for the argument to be a document of any type.)
- *documentShouldClose*: called when the application is preparing to close a document.
- *documentDidClose*: called when the application has closed a document.
- *documentShouldSave*: called when the application is trying to determine whether a given document should be saved.
- *documentWillSave*: called when the application is about to begin saving a document. (note that this will only be called after a successful return from a ‘documentShouldSave’.
- *documentDidSave*: called after a document has been saved successfully.
- *documentWillUnlock*: called when BBEdit is going to make a document writeable. (For example, when you click the pencil to unlock a document)
- *documentDidUnlock*: called when BBEdit has successfully made a document writeable.
- *documentWillLock*: called when BBEdit is going to make a document read-only.
- *documentDidLock*: called when BBEdit has successfully made a document read-only.

Using Attachment Scripts

Scripts attached to events must be stored in the "Attachment Scripts" folder of BBEdit's application support folder (see page 33).

You can write one script to handle each attachment point, or one script to handle the attachment points for an entire class of objects, or one script to handle all of the attachment points for the entire application.

You can also mix and match scripts to meet specialized needs: for instance, by using one script to implement a particular attachment point for documents, and a second script to handle the remaining attachment points.

BBEdit associates scripts to attachment points by means of the script's file name. There are three ways to specify a script's role:

1. **<ObjectClass>.<entryPoint>**
2. **<ObjectClass>**
3. **<ApplicationName>**

The first form is the most specific: the 'ObjectClass' may be either "Application" or "Document", while the 'entryPoint' is one of the attachment points described above appropriate to that object class.

For example, a script which implemented only the *documentDidSave* attachment point should have the name "Document.documentDidSave.scpt" and contain a subroutine named 'documentDidSave', thus:

```
on documentDidSave(myDoc)

    -- do something useful and appropriate

end documentDidSave
```

Note Adding the filename suffix '.scpt' is not mandatory, but you should follow the current system conventions suggested when creating scripts with the AppleScript Editor (or any other script editor such as Script Debugger).

The second form allows you to implement all of the attachment points for a single object class in a single script file, if desired.

For example, you could create a script named "Application.scpt" containing subroutines for as many of the application attachment points as you wish:

```
on applicationDidFinishLaunching

    -- do something relevant

end applicationDidFinishLaunching

on applicationShouldQuit

    -- hello world

    return (current date as string contains "day")

end applicationShouldQuit
```

Likewise, to implement all of the attachment points for the Document class, you could create a script named “Document.scpt”, and put subroutines in it for the document attachment points:

```
on documentDidSave

    -- do something relevant

end documentDidSave

    ...

on documentWillClose

    ...

end documentWillClose
```

The third form allows you to write a single all-encompassing script which contains subroutines for all of the attachment points in the application. To do this, name the script “BBEdit.scpt” and include whatever subroutines you wish to implement. For example:

```
on applicationShouldQuit

    -- hello world

    return (current date as string contains “day”)

end applicationShouldQuit

on documentWillClose

    ...

end documentWillClose
```

When figuring out which script to run, BBEdit will first look for a script whose name exactly matches the attachment point, e.g. “Document.documentShouldSave.scpt”. If there is no such script, BBEdit will then look for a script whose name matches the object class at the attachment point, e.g. “Document.scpt”. Finally, if there are no scripts with either an exact or a class match, BBEdit will look for an application-wide script: “BBEdit.scpt”.

Note You do **not** have to implement attachment subroutines for all attachment points, or for all classes—only the ones you need. If there is no attachment script or subroutine, BBEdit proceeds normally.

Using an Attachment Script to Perform Authenticated Saves

BBEdit supports a special attachment point for the Document class: *documentShouldFinalizeAuthenticatedSave*. This attachment point will be called whenever an authenticated save is necessary (for text documents only).

The following sample script illustrates how to use this facility (the comments are important, so please read them!):

```
on documentShouldFinalizeAuthenticatedSave(theDocument,
tempFilePath, destinationPath)

    -- on input: tempFilePath points to the contents
    -- of the document written to a temp file, ready
    -- to move to the destination; destinationPath is
    -- where the file should be copied.

    -- on exit: if the operation succeeded, delete the
    -- temp file (or else the application will assume
    -- the operation failed) and return YES for success

    -- this is pretty straightforward:
    --   "cp tmpFilePath destinationPath"

    do shell script "cp" & " " & quoted form of tempFilePath
& " " & quoted form of destinationPath with administrator
privileges

    -- now remove the temp file, this indicates to
    -- the application that we did the work

    do shell script "rm" & " " & quoted form of tempFilePath

    return true

end documentShouldFinalizeAuthenticatedSave
```

Filtering Text with AppleScripts

The Text Filters folder in BBEdit's application support folder contains executable items, such as compiled AppleScripts, Automator workflows, and Unix filters, which you may apply to the active document via the Apply Text Filter submenu of the Text menu.

When you apply such an item, BBEdit will pass either the selected text (or the contents of the active document, if there is no selection) as a reference to a 'RunFromBBEdit' entry point within your AppleScript, and your script should return a string which BBEdit will use to replace the selected text (or the contents of the document). If your script does not contain a 'RunFromBBEdit' entry point, BBEdit will call its run handler, again passing a reference to the current selection range.

BBEdit's Scripting Model

This section provides a high-level overview of BBEdit's scripting model that will, where appropriate, contrast the current scripting framework against older versions of BBEdit, and suggest how you can modify your existing scripts for compatibility.

IMPORTANT Because BBEdit's scripting dictionary changes whenever we add features, it should be considered the definitive reference in any situation where it and this document differ. We have found Script Debugger from Late Night Software to be an excellent tool for browsing and navigating BBEdit's scripting dictionary, as well as for preparing and testing scripts.

```
https://www.latenightsw.com/
```

Script Compatibility

Since BBEdit's scripting model has changed over time, scripts prepared for much older versions may need revision in order to work properly. For example, since BBEdit allows multiple documents to be open within a single text window, you may need to revise existing scripts which presume documents and windows are identical.

Distinguishing Between Script Elements

Because different applications handle different types of data, you should be aware that the actual data, or the interface items, referred to by a particular name may not be consistent from application to application. The following sections describe how several common elements are handled in BBEdit.

Applying Commands to Text

Since BBEdit supports opening multiple documents within a single text window, all scripting commands which operate on text must specifically target the text contents of a window, or a document within that window, rather than the window itself.

For example, you may use:

```
count lines of text of text window 1
```

or:

```
count lines of active document of text window 1
```

but not:

```
count lines of window 1
```

Documents vs. Windows

In substantially older versions of BBEdit, the object classes *document* and *window* could be used interchangeably, and generally had the same properties listed in the scripting dictionary. This is no longer the case.

The class *window* corresponds to a window (of any type—text or otherwise) on screen, and thus the properties of the *window* class refer strictly to properties of a window on screen. If a document is associated with a window, the document is accessed as the *document* property of the window:

```
document of text window 1
```

The class *document* refers to a document, and as with a window, the document's properties pertain strictly to the condition of a document (that is, something that can be saved to disk and opened later). Note that this does not mean a document must be saved to a file, only that it could be.

As a rule, documents and windows are associated with each other, but it is important to remember that there is not a one-to-one correspondence between windows and documents. For example, the About box is a window which has no document associated with it. Furthermore, in current versions of the application, there is no such thing as a document with no associated window.

Here is a general overview of the object classes used in BBEdit:

Classes of Windows

- *window*: the basic window class contains properties that can be fetched and set for any window on screen: position, size, and so forth.
- *palette*: the palette class refers to windows that float above all others on the screen; the HTML tools palette, scripts list, and so on.
- *text window*: the text window class provides properties which are specific to text-editing windows as on-screen entities. These properties pertain mostly to the display of text in the window: *show invisibles*, *auto_indent*, and so on. In addition to the text-editing-specific properties, the basic window properties are also accessible.
- *project window*: provides a way to reference windows corresponding to open projects. A group window does not present any properties beyond the basic *window* class, but provides a way to differentiate project windows from other types of window.
- *disk browser window*: provides a way to reference windows corresponding to open disk browsers. A disk browser window does not present any properties beyond the basic *window* class, but provides a way to differentiate disk browser windows from other types of window.
- *results browser*: provides a way to reference results generated by a batch operation. A results browser does not present any properties beyond the basic *window* class, but provides a way to differentiate results windows from other types of window.
- *search results browser*: a subclass of results browser, referringspecifically to the results of a single-file Find All command or a multi-file search.

Classes of Document

As with windows, there are various classes of document:

- *document*: the basic document class contains properties that apply to any sort of document: whether it has unsaved changes, the alias to the file on disk, and so on.
- *text document*: text documents contain information specific to text files opened for editing in BBEdit.

- *group document*: refers to a document corresponding to an open project. A project document does not present any properties beyond the basic *document* class, but provides a way to differentiate project documents from other types of document.
- *picture document*: refers to a document corresponding to an open picture file. A picture document does not present any properties beyond the basic *document* class, but provides a way to differentiate picture documents from other types of document.
- *movie document*: refers to a document corresponding to an open QuickTime movie file. A movie document does not present any properties beyond the basic "document" class, but provides a way to differentiate movie documents from other types of document.
- *QuickTime document*: refers to a document corresponding to an imported Quicktime image file. A QuickTime document does not present any properties beyond the basic "document" class, but provides a way to differentiate QuickTime documents from other types of documents.

"Lines" and "Display_lines"

The "line" element refers to a "hard" line, that is, a stream of characters that begins at the start of file or after a line break, and which ends at the end of file or immediately before a line break. This is consistent with the semantics of "line" in hard-wrapped documents, and these semantics also apply within soft-wrapped documents.

The "display_line" element refers to a line of text as displayed on screen (bounded by soft and/or hard line breaks).

The "startLine" and "endLine" properties of a text object always refer to the "hard" start and end of lines. In other words, if a text object crosses multiple soft-wrapped lines, the startLine and endLine properties will be the same.

Both "startDisplayLine" and "endDisplayLine" properties are part of the text object class. These serve the same purpose as the startLine and endLine semantics for soft-wrapped views in older versions of BBEdit.

Getting and Setting Properties

One significant improvement in BBEdit's new scripting framework is the ability to get and set multiple properties of an object with a single scripting command. Every object has a property called *properties*. This property returns a record which contains all of the properties which can be fetched for that object. For example, the script command

```
properties of text window 1
```

will return a result like this one:

```
{{id:55632400, container:application "BBEdit", bounds:{31, 44,
543, 964}, closeable:true, collapsed:false, index:1,
modal:false, file:alias "Hard
Disk:Users:Shared:doc_examples:index copy.html", modified:false,
name:"index copy.html", position:{31, 44}, resizable:true,
selection:"", contents:"..."}
```

Conversely, to set one or more properties at once is very easy:

```
set properties of text window 1 to { show invisibles: true, show
spaces : true, soft wrap text : true }
```

Only the properties specified will be changed. The rest will not be modified.

It is important to note that when setting properties in this fashion, you can only set modifiable properties. If you attempt to set any read-only properties, a scripting error will result:

```
set properties of text window 1 to { show invisibles: true,
modal: false, expand tabs: true }
```

The above script command will turn on Show Invisibles and then report a scripting error, since *modal* is a read-only property.

Performing Actions

The following sections provide basic information on how to perform various common actions via AppleScript.

Scripting Searches

The ability to script searches presents you with a very powerful tool, since you can prepare a script which instructs BBEdit to perform a whole series of search or search and replace operations.

Consider the scripting command below:

```
tell application "BBEdit"

find "BBEdit(.+)$" searching in document of text window 1 ¬
  options { search mode: Grep } with selecting match

end tell
```

In substantially older versions, the *find* command always operated on the front window; however, you must now explicitly specify the text to be searched, either by specifying an explicit tell target, or by supplying a *searching in* parameter. So the following scripts are equivalent:

```
tell application "BBEdit"
    find "BBEdit" searching in document of text window 1
end tell
```

and

```
tell application "BBEdit"
    tell document of text window 1
        find "BBEdit"
    end tell
end tell
```

Note that either the tell-target or the *searching in* parameter must resolve to something that contains text. As a shortcut, you can specify a window, and if the window contains text, the search can proceed. You can also specify a text object:

```
find "Search Text" searching in (lines 3 thru 5 of document of
text window 2)
```

Please also bear in mind that the defaults for parameters not specified in the *find* command are independent of those visible within the user interface (that is, the Find and/or Multi-File Search windows).

When performing a *find*, BBEdit will return a record describing the results of the search. This record contains a Boolean which indicates whether the search was successful, a reference to the text matched by the search, and the text string matched by the search. Given the first example above, the results might look like this (after reformatting for clarity):

```
{found:true,
found object:characters 55 thru 60 of text window 1 of
application "BBEdit",
found text:"BBEdit"}
```

Scripting Single Replaces

To do a single find and replace via AppleScript, you can write:

```
tell application "BBEdit"

set result to (find "BBEdit" searching in text of ¬
    text window 1 with selecting match)

    if (found of result) then
       set text of (found object of result) to "Replacement"
    end if

end tell
```

When performing a grep search, you cannot just replace the matched pattern with a replacement string; the grep subsystem needs to compute the substitutions. The *grep substitution* event is provided for this purpose; given a preceding successful Grep search, it will return the appropriate replacement string. So if you perform a grep search, the script would look like:

```
tell application "BBEdit"

set result to find "BBEdit(.+)$" searching in text of ¬
    text window 1 options {search mode:grep}

    if (found of result) then
         set text of (found object of result) to ¬
             grep substitution of "\\1"
    end if

end tell
```

Note that when using a backslash "\" character in AppleScript, it needs to be "escaped" by means of another backslash; thus, in the above example, "\\1" used in the script, will become the grep replacement string "\1" when passed to BBEdit.

Scripting Multi-File Searches

In BBEdit, a multi-file search is a simple extension of the *find* scripting command. To search a single file or folder for all occurrences matching the search parameters, specify the file or folder as the *searching in* parameter of the search.

For example, to find all occurrences of "index.html" in a website, one might use the following scripting command:

```
find "index.html" searching in (alias "Files:WebSite:")
```

Likewise, to find JavaScript line comments:

```
find "//.+$" searching in (alias "Files:WebSite:") ¬
    options {search mode: Grep}
```

To search in a single file:

```
find "crash" searching in (alias "Files:WebSite:index.html")
```

Scripting the Clipboard

BBEdit has multiple clipboards. These are fully accessible via the scripting interface. Due to operating system constraints, most clipboard operations require BBEdit to be frontmost.

Here are some examples:

```
count clipboard
```

- Returns the number of clipboards supported by the application

```
clipboard 1
```

- Returns {index:1, contents:"Files:WebSite:", length:14, is multibyte:false, display font:"ProFont", display font size:9, style:{plain}}

```
clipboard 1 as text
```

- Returns "Files:WebSite:"

```
clipboard 1 as reference
```

- Returns clipboard 1 of application "BBEdit"

```
current clipboard
```

- Returns the current clipboard as a record (you can coerce it to reference or text or get individual properties)

To set the text in a given clipboard to literal text:

```
set contents of clipboard 3 to "foobar"
```

To set the text in a clipboard to text represented by an object specifier:

```
set contents of clipboard 3 to selection of window 2
```

To copy the contents of one clipboard to another:

```
set contents of clipboard 5 to clipboard 3
```

or, to set the current clipboard to the contents of a different clipboard, (thus making it exportable to the system clipboard):

```
set current clipboard to clipboard 3 as text
```

or finally, with even less typing involved:

```
set current clipboard to clipboard 5
```

To make any clipboard the current clipboard, select it:

```
select clipboard 5
```

Scripting Text Factories

You can apply a text factory to a file via the AppleScript interface. The minimum invocation is:

```
apply text factory <file reference> to <reference>
```

The "to" parameter can be a single reference or a list of references, as for the multi-file "find" or "replace" events.

Optional parameters include "filter", "saving", "recursion", "text files only", "search invisible folders", all with the same meanings as in the multi-file "replace" event.

Setting Text Encodings

When specifying the encoding to use for opening or saving a file, you may either use the encoding's internet name, or its exact display name (as shown in the Read As popup menu).

For example:

```
open {file "Hard Disk:Users:Shared:example.txt"} reading as
"Western (ISO Latin 1)"

open {file "Hard Disk:Users:Shared:example.txt"} reading as
"iso-8859-1"
```

Arranging Documents and Windows

BBEdit provides considerable control for handling windows and documents both directly and via AppleScript.

Opening Documents

The "open" command supports additional options, which allow you to override your window handling preferences on a case by case basis:

```
open aFileList opening in <value>
```

As in previous releases, <value> may be a reference to an existing text window. However, you may instead specify "front_window", "new_window", or "separate_windows", which have the following effect:

- front_window: All files in aFileList are opened in the frontmost text window. (If there is no text window open, BBEdit will create a new one.)

- new_window: All files in aFileList are opened into a new text window.
- separate_windows: Each file in aFileList is opened into its own text window.

Moving Documents

The “move”command can be used to move text documents between text windows. For example:

```
tell application "TextWrangler"
    if (count of text windows) > 0 then
        select text window 1
        repeat while (count of text windows) > 1
            set ct to count documents of text window 2
            repeat with i from 1 to ct
                move document 1 of text window 2 to text window 1
            end repeat
        end repeat
    else
        beep
    end if
end tell
```

Referencing Documents

Previously, documents were indexed inside of multi-document windows by their display order in the sidebar. This meant that “document 1” of the application might not be the active document, which in turn required scripts to make special provisions to deal with the presence of multiple documents in a single window.

In order to handle this, BBEdit 8.0 introduced the “active document” property, which you could always use to specify the currently active document of a given text window. For example:

```
active document of text window 1 of application “BBEdit”
```

Although BBEdit still supports the “active document” property, this is no longer necessary. Instead, if a text window is frontmost, the following references:

```
document 1 of application "BBEdit"
```

```
document 1 of text window 1 of application "BBEdit"
```

```
active document of text window 1 of application "BBEdit"
```

all resolve to the same document. The side effect of this change is that if you wish to access documents within a text window by index, that index is:

a) not related to the visual ordering of documents in the sidebar, and,

b) documents’ indexes may change over time

This situation is effectively no different than handling documents which are contained in individual text windows, i.e. the index will change over time when you select different windows. If your script needs to keep a permanent references to a particular document, you should refer to that document by its id rather than its index.

Common AppleScript Pitfalls

Here are some things to watch out for when scripting BBEdit with AppleScript.

The Escape Issue

AppleScript uses the backslash character as an escape character. You can use \n or \r to specify a literal line break or \t to indicate a tab character. More importantly, you can use \" or \' to include a quote mark or apostrophe in a string that is delimited by quotes or apostrophes. If you want to specify a literal backslash, you must write \\ i.e. a pair of backslashes.

That's not all that confusing until you start writing AppleScripts that call on BBEdit's powerful grep searching capability. BBEdit *also* uses the backslash as an escape character. If you want to search for an actual backslash in a document, you have to tell BBEdit to search for \\. However, if you do that in AppleScript, you must keep in mind that AppleScript will first interpret the backslashes before passing them to BBEdit. To pass one backslash to BBEdit from AppleScript, you must write two in AppleScript.

So to tell BBEdit to search for a single literal backslash from an AppleScript, you must write no fewer than *four* backslashes in the script. Each pair of backslashes is interpreted as a single backslash by AppleScript, which then passes two backslashes to BBEdit. And BBEdit interprets those two backslashes as a single one for search purposes. (This proliferation of backslashes can make your scripts look a bit like a blown-over picket fence.)

The Every Item Issue

When writing a script that loops through every item of a BBEdit object (for example, every line of a document), do not do it like this:

```
repeat with i in every line of text document 1
    -- do stuff here...
end repeat
```

This forces BBEdit to evaluate "every line of document 1" every time through the loop, which will slow your script significantly. Instead, write

```
set theLines to every line of text document 1
repeat with i in theLines
    -- do stuff here...
end repeat
```